

Как в 1С оптимизировать запрос – пример ускорения в 6 раз

Часто при внедрениях программ «1С: Предприятие 8» возникают моменты, в которых, казалось бы, обычные запросы работают достаточно медленно, несмотря на свою простоту. Попробуем показать пути оптимизации таких запросов. Нам необходим запрос, работающий шесть и более секунд. К счастью, нашелся подходящий кандидат, причем из реального проекта:

ВЫБРАТЬ

```
ДенежныеСредстваКПоступлениюБезналичныеОбороты.Документ.Контрагент.Партнер,  
МЕСЯЦ (ДенежныеСредстваКПоступлениюБезналичныеОбороты.Период) КАК Месяц,  
ГОД (ДенежныеСредстваКПоступлениюБезналичныеОбороты.Период) КАК Год,  
СУММА (ДенежныеСредстваКПоступлениюБезналичныеОбороты.СуммаПриход) КАК СуммаПриход  
ИЗ  
РегистрНакопления.ДенежныеСредстваКПоступлениюБезналичные.Обороты(, , Месяц,  
Документ.Контрагент.Партнер = &Партнер) КАК  
ДенежныеСредстваКПоступлениюБезналичныеОбороты
```

СГРУППИРОВАТЬ ПО

```
ДенежныеСредстваКПоступлениюБезналичныеОбороты.Документ.Контрагент.Партнер,  
МЕСЯЦ (ДенежныеСредстваКПоступлениюБезналичныеОбороты.Период) ,  
ГОД (ДенежныеСредстваКПоступлениюБезналичныеОбороты.Период)
```

С первого взгляда все хорошо, но опытный программист увидит неоптимальный код в нашем запросе. Источником проблем выступает параметр виртуальной таблицы, а точнее – обращение через «две точки» в фильтре.

Согласитесь, в общем случае, ничего ужасного и запретного тут нет, но есть одно «но»: поле «Документ» имеет составной тип, и при получении реквизитов данного поля выполняется соединение с каждой таблицей, входящей в составной тип этого поля.

Самым первым вариантом решения в голову приходит использовать конструкцию языка запросов «ВЫРАЗИТЬ», дабы привести наше поле к некоторому определенному типу, что избавит нас от соединения с лишними таблицами. Но по ряду ограничений и условий данный путь не подходит:

1. Нам нужны все документы, содержащиеся в составном типе. Таковы условия постановки задачи. Получается, что необходимо фильтровать все типы документов, входящие в составной тип.
2. Даже если бы не было предыдущего ограничения, то обращение через «две точки» никуда не делось.
3. Если бы можно было использовать «ВЫРАЗИТЬ», то это не спасало бы ситуацию: в параметрах виртуальной таблицы «ВЫРАЗИТЬ» не дает прироста производительности.

Оптимизация

Вывод: прежде всего, необходимо избавиться от обращения через «две точки», и при этом не испортить саму логику нашего запроса.

Из нескольких вариантов решения задачи предлагаем два следующих варианта:

Вариант 1

В регистр «ДенежныеСредстваКПоступлениюБезналичные» добавить новое измерение «Партнер», заполняя его при записи движений документов. Ввиду использования условия по данному измерению его необходимо проиндексировать.

После внесенных нами изменений у нас достаточно легко получится наложение фильтра на новое измерение в параметрах виртуальной таблицы:

```
РегистрНакопления.ДенежныеСредстваКПоступлениюБезналичные.Обороты(, , Месяц,  
Партнер = &Партнер) КАК ДенежныеСредстваКПоступлениюБезналичныеОбороты
```

Что мы видим? Этот запрос начинает работать моментально. И это, к сожалению, единственный положительный момент, минусов наблюдается существенно больше:

Главный минус – изменение структуры конфигурации, возникают проблемы при последующих обновлениях, использовании типовых обменов и т.д.

К тому же, у нас хранится дублируемая информация, что приводит к увеличению размера таблицы, а установка признака индексирования повышает скорость чтения, но при этом замедляет запись в регистр.

Поэтому, рассмотрим второй вариант.

Вариант 2

Можно попробовать изменить запрос так, чтобы фильтр по полю «Документ» накладывался примерно следующим образом:

```
РегистрНакопления.ДенежныеСредстваКПоступлениюБезналичные.Обороты(, , Месяц, Документ  
В (ВЫБРАТЬ Ссылка ИЗ ВТ_ДокументыСПартнером)) КАК  
ДенежныеСредстваКПоступлениюБезналичныеОбороты
```

Что необходимо сделать, чтобы наш запрос пришел к подобному виду?

Вначале соберем все документы, входящие в составной тип поля «Документы». Для них должно соблюдаться условие: `Контрагент.Партнер = &Партнер`

В нашем составном типе определены пять документов, причем искомый реквизит «Контрагент» присутствует в четырех документах: `ПоступлениеБезналичныхДенежныхСредств`, `СписаниеБезналичныхДенежныхСредств`, `РасходныйКассовыйОрдер`, `ОперацияПоПлатежнойКарте`.

Далее сформируем временную таблицу для фильтрации. В ней будут документы, у которых реквизит «Партнер» равен нужному значению. Применим полученный фильтр по документам в нашем запросе:

```
ВЫБРАТЬ ОперацияПоПлатежнойКарте.Ссылка
ПОМЕСТИТЬ ВТ_ДокументыСПартнером
ИЗ Документ.ОперацияПоПлатежнойКарте КАК ОперацияПоПлатежнойКарте
ГДЕ ОперацияПоПлатежнойКарте.Контрагент.Партнер = &Партнер
```

ОБЪЕДИНИТЬ ВСЕ

```
ВЫБРАТЬ ПоступлениеБезналичныхДенежныхСредств.Ссылка
ИЗ Документ.ПоступлениеБезналичныхДенежныхСредств КАК
ПоступлениеБезналичныхДенежныхСредств
ГДЕ ПоступлениеБезналичныхДенежныхСредств.Контрагент.Партнер = &Партнер
```

ОБЪЕДИНИТЬ ВСЕ

```
ВЫБРАТЬ РасходныйКассовыйОрдер.Ссылка
ИЗ Документ.РасходныйКассовыйОрдер КАК РасходныйКассовыйОрдер
ГДЕ РасходныйКассовыйОрдер.Контрагент.Партнер = &Партнер
```

ОБЪЕДИНИТЬ ВСЕ

```
ВЫБРАТЬ СписаниеБезналичныхДенежныхСредств.Ссылка
ИЗ Документ.СписаниеБезналичныхДенежныхСредств КАК СписаниеБезналичныхДенежныхСредств
ГДЕ СписаниеБезналичныхДенежныхСредств.Контрагент.Партнер = &Партнер
;
```

////////////////////////////////////

```
ВЫБРАТЬ
&Партнер,
МЕСЯЦ (ДенежныеСредстваКПоступлениюБезналичныеОбороты.Период) КАК Месяц,
ГОД (ДенежныеСредстваКПоступлениюБезналичныеОбороты.Период) КАК Год,
ДенежныеСредстваКПоступлениюБезналичныеОбороты.СуммаПриход КАК СуммаПриход
ИЗ
РегистрНакопления.ДенежныеСредстваКПоступлениюБезналичные.Обороты (, , Месяц, Документ
В
(ВЫБРАТЬ
ВТ_ДокументыСПартнером.Ссылка
ИЗ
ВТ_ДокументыСПартнером) ) КАК ДенежныеСредстваКПоступлениюБезналичныеОбороты
```

С другой стороны, можно сначала получить контрагентов с данным партнером и затем искать документы с фильтром по контрагенту, но особой разницы в скорости не наблюдается.

После проведенной оптимизации запрос стал выполняться менее одной секунды!

Да, при этом он стал сложнее, но нет необходимости в изменении структуры метаданных, как в первом варианте.

Резюме

Вам представлен вариант решения оптимизации достаточно простого запроса, при котором не возникло необходимости в перестройке метаданных, создания дополнительных индексов.

Рекомендуем стараться провести оптимизацию способом, приведенным во втором варианте, посредством переписывания текста самого запроса.

Что делать, если 1С тормозит, зависает и вылетает?



Если Вам понравились эти материалы, мы приглашаем Вас пройти предварительную регистрацию на тренинг «Оптимизация производительности 1С:Предприятие 8 и подготовка к 1С:Эксперт» по адресу: <http://kursy-po-1c.ru/optimize1C>

Чему Вы научитесь после прохождения курса:

- самостоятельно решать проблемы производительности
- проводить анализ системы и выявлять «узкие места» в плане производительности
- находить медленные запросы, наиболее сильно влияющие на систему, и оптимизировать их
- читать и понимать план запроса
- выявлять почему данная конкретная операция выполняется медленно
- оценивать загруженность оборудования
- выявлять и решать проблемы избыточных блокировок
- выявлять и решать проблемы взаимных блокировок
- работать с ЦУП и Тест-Центр
- работать с облачными сервисами контроля производительности
- распараллеливанию кода на 1С
- расследовать и решать проблемы стабильности
- настраивать кластер серверов наиболее оптимальным для производительности образом
- настраивать отказоустойчивый кластер серверов 1С
- настраивать и использовать технологический журнал для решения проблем производительности и стабильности